
ABSTRACT 
This paper examines the SAS® Output Delivery System (ODS) Report Writing Interface 
(RWI) to streamline the publication process of "Health, United States", an annual report on 
health and health care in the United States. The report’s tables contain multiple nested 
categories within rows and columns and are suitable for printing (PDF) and downloading 
(Microsoft Excel). The publication process from SAS datasets currently involves several 
labor-intensive steps. In addition, Section 508 of the Rehabilitation Act of 1973 requires all 
federal agency website content to be accessible to people with disabilities. The process of 
manually labeling content for screen readers (“tagging”) presents an additional challenge for 
these complex nested tables. 

The publishing process is being automated and the output delivery is being enhanced using 
the RWI.  The ODS templates and Cascading Style Sheets are used to standardize the 
format of the output and varied formatting using conditional processing. This approach 
maintains the existing output formats while limiting the manual processing, decreasing 
production time, and increasing quality. This approach allows automated production of a 
variety of file formats including PDF, HTML5, and Excel. Using the RWI and SAS accessibility 
features, most of the manual Section 508 compliance task of the output should be 
eliminated. This paper will highlight these proposed changes for production of “Health, 
United States 2017.” 

INTRODUCTION 
Health, United States (https://www.cdc.gov/nchs/hus/index.htm) is an annual report about 
the health status of people living in the United States. It is submitted by the Secretary of 
the United States Department of Health and Human Services (HHS) to the President and the 
Congress. The report is prepared by the principal health statistics agency of the federal 
government, the National Center for Health Statistics (NCHS).  

The Health, United States 2018 Chartbook compiles multiple topics of public interest and 
policy relevance. The 47 supplementary online-only trend tables are complex, involving 
multiple nested categorical values within rows and columns. Producing this report is a 
complex effort and involves multiple levels of review and scrutiny, with iterative manual 
proofing and quality control checks. Automating the production of the trend tables would 
result in a more streamlined process with improved accuracy.   

Health, United States trend tables are available online as Microsoft (MS) Excel tables and as 
individual PDF documents that must be 508 compliant. Section 508 is an amendment to the 
Rehabilitation Act of 1973 that requires Federal agencies to make their Electronic and 
Information Technology (EIT) accessible to people with disabilities 
(https://www.section508.gov/).     

This paper presents a case study using the SAS Output Delivery System (ODS) Report 
Writing Interface (RWI) to streamline the production process of the Health, United States 
trend tables. The 2017 Health, United States “Table 1. Resident population, by age, sex, 

https://www.cdc.gov/nchs/hus/index.htm
https://www.section508.gov/

race, and Hispanic origin: United States, selected years 1950–2016” was used to 
demonstrate these capabilities. ODS templates, Cascading Style Sheets (CSSs), in-line 
formatting, and Unicode characters were used to duplicate the formatting of the existing 
PDF trend table documents. The RWI DATA _NULL_ table generation methods were used to 
create the trend tables with the nested row and column headers, footnote reference marks, 
data placement, and conditional formatting. Using the same RWI code, the trend tables can 
be output in various file formats with minimal code changes. The generation of the existing 
PDF files as well as HTML5 as a replacement for the existing Excel file format are 
demonstrated. 

The results of this case study using the SAS RWI with various formatting and accessibility 
features are expected to streamline production processes while ensuring accuracy of the 
overall output.       

HEALTH, UNITED STATES TREND TABLES – CURRENT 
PRODUCTION 
Health, United States is currently being redesigned, which creates opportunities to update 
the process for generating the trend tables. The current production process involves several 
manual steps. The team is looking into new ways to produce these tables in a more 
automated way. While maintaining the 508 compliant PDF files of the past, the team is also 
interested in providing the data in more user-friendly digital formats for screen readability 
and ease of data download and import to various software. 

Dynamic Data Exchange (DDE) is a key component of the current trend table generation 
process. DDE is a Microsoft (MS) method of inter-process communication, introduced for 
Windows in 1987, allowing SAS to communicate with MS Excel. DDE is the method currently 
used in the production of the Health, United States trend tables to output data from SAS 
datasets into preformatted Excel spreadsheets (table “shells”). These spreadsheets are later 
used to generate the PDF files for publication. DDE continues to work even in modern 
versions of Windows but has been superseded by newer technologies. When updates to 
related software are applied, DDE sometimes requires coding or configuration modifications.    

The current process of generating annual updates to trend tables in Health, United States 
that is being redesigned involves the following: 

1. Create SAS dataset of estimates for latest year of data using SAS and SAS-callable 
SUDAAN  

2. Update an existing MS Excel spreadsheet to add blank rows and columns to accept 
new data values 

3. Update SAS programs using DDE to point to appropriate rows and columns in Excel 
file to output the new data 

4. Review the file manually to confirm that the appropriate values are entered in the 
appropriate table cells 

5. Create PDF documents from the Excel file using third-party software 

6. Manually review and tag the PDF to be Section 508 compliant  

Publish the PDF file and Excel format fileThe Health, United States Trend Tables are 
currently published to the Health, United States website in 2 formats: 

• Excel files that include all existing / historical data available. These are designed for 
readability on a screen but are not suitable for 8 ½ x 11” printing. 

• PDF files that include a subset of the data, with only some years selected for 
printing. These are suitable for 8 ½ x 11” printing.  

HEALTH, UNITED STATES TREND TABLE (PDF FORMAT). 

 
Figure 1. Sample of Existing PDF Trend Table (Sections of Table 1) 

 

As seen in Figure 1, the current Table 1 PDF version of Health, United States has 

• A data table with multiple levels of row and column headers (1) 

• A hyperlink to the website where both Excel and PDF files are available (2) 

• A footnotes section at the bottom of the last page with citations embedded with data 
(3) 

All these components of the table involve varying fonts (style & size) as well as custom 
header, footer, and data placement. 

  



5 

HEALTH, UNITED STATES TREND TABLE (EXCEL FORMAT)  

 
Figure 2. Sample of Existing Excel Trend Table (Sections of Table 1) 

 

As seen in Figure 2, the current Excel version of Table 1 has:   

• The same table structure as the PDF 

• A footnotes section at the bottom of the last page with citations embedded with data 
(1) 

• Custom header and data placement with various types of formatting 

• Frozen header rows (2) 

The Excel version of the trend tables are not in a “flat file” format of rows and columns, 
which would be ideal for data import to various software. Some tables contain additional 
years of data and standard errors not included in the PDF (3) 

HEALTH, UNITED STATES TREND TABLES: REVISED PROCESS 
This section presents a case study using the SAS Output Delivery System (ODS) Report 
Writing Interface (RWI) to streamline the production process of the Health, United States 
trend tables. Using the SAS RWI reduces much of the manual review and editing due to the 
automation it provides. Various output files including; PDF, HTML5, and Excel, referred to as 
ODS Destinations, can be created using the same code. There is no longer a need to use 
DDE, which is an outdated technology and no longer supported for use by SAS, for data 
transfer from SAS to Excel. The third-party software is no longer needed in this revised 
process to produce the PDF file from the Excel source file. The new process takes the data 
stored in SAS datasets and directly outputs the needed trend table output files. Lastly, the 
RWI provides an automated 508 compliance process which reduces the manual 508 
compliance process. 

CHANGES TO FILE TYPES 
The new process would replicate the PDF produced by the current process and would create 
an HTML5 file instead of Excel, to increase screen readability and responsiveness to the 
tables accessed via a tablet or phone. Although not created through the use of the RWI, a 
comma-separated format (CSV) file would be produced as well to provide an open format 
file that can be easily downloaded by researchers wishing to use the trend table data as a 
source for further analysis using various software packages (e.g., R).   

 

NEW PROCESS FOR ANNUAL REPORT GENERATION FOR EACH OF THE 
TREND TABLES 

1. Create data file of estimates for latest year and append to the “Master” SAS dataset 
(containing historical data) for each trend table 

2. Run the SAS program to generate the output reports: PDF & HTML5 (or Excel) 

3. Output the “Master” SAS dataset into a CSV data file  

4. Review the files for data, style, formatting, and 508 compliance   

USING SAS TO CREATE THE CUSTOM REPORTS 

SOURCE DATA 
To output the source data in the desired format and order, for the new process, the data will 
need to be stored appropriately. 

   
Figure 3. Sample of Headers and Data Values to Load for Final Trend Table 1 

  

The source data is stored in a SAS dataset. To 
make the various data group headers such as “All 
persons” or “Under 1 year” as seen in Figure 3 
appear in the correct order; data headers and 
labels are stored as numbers and formats are 
used for the labels. “All persons” is listed first so 
it is stored as FORMAT_NAME: 
GROUP_NUM_001=0 as seen in Table 1 The 
formats, see Table 2, are then applied. This 
causes them to be listed in the correct order 
when sorted for the final table output. Table 3 
shows a sample of the formatted data that will be 
used for the final output tables. 

 
Table 1. Master Data (Unformatted 
Data Values) 

    
Table 2. Sample of Formats 

 
Table 3. Master Data (with SAS 
Formats) 

 

The PDF file contains fewer years of data, so the code includes a conditional section for the 
PDF file where those additional data rows are dropped from a work SAS dataset 

STYLE CONTROL 
To properly format the new files, to match or enhance the existing files, SAS ODS and the 
RWI includes many different options for formatting, all of which can be applied to all the 
output destinations including PDF, HTML5, and Excel. Some of these formatting options 
including: Unicode characters, in-line formatting, and Cascading Style Sheets (CSSs) with 
ODS templates, will be demonstrated below. Using the RWI, the tables can be designed with 
these various style controls and applied to all the file types with minimal additional coding. 

Unicode Characters 
Health, United States relies on footnotes throughout the trend tables to explain important 
details of the data and headers. If the footnotes are just stylized as superscripts, then 
copying and pasting the data may result in the superscript format being dropped and 
appearing as part of the numeric value. Instead, inputting the appropriate Unicode 
characters to represent the superscripted numbers preserves the formatting as an actual 
symbol character. 

 
Figure 4. Sample NEW PDF Demonstrates 
Unicode Characters 

 

 

Conditionally output Unicode Superscript 
and color data value Red if the Footnote 
flag is set as seen in Figure 4. 

 

 

The following code illustrates the use of formatting options including the Unicode (00B9) 
which is a superscript 1: 

 
obj.format_cell(data:"^{unicode 00B9} "|| 

                        trim(left(put(age_cat_02,comma8.))),  

                          style:'DataCells', just:'right', 

                          style_attr:"color=red") ; 

  

NOTE: using ^{super 1} instead of the Unicode returns a formatted superscript 1, not the 
Unicode character 

 

ODS In-line Formatting 

 
Figure 5. Top of SAS Generated PDF Report for Table 1 of Health, United States 

The header of the Table 1 report for Health, United States shown in Figure 5 demonstrates a 
few examples of in-line formatting. To use ODS in-line formatting, the first step is to assign 
an “escape character” which identifies the inline formatting symbol. This means the SAS 
compiler identifies the code following the escape character as in-line formatting code. The 

user should select a seldom-used character to serve as an escape character. In this case the 
“caret” character “^” is used: 
 

ODS escaprchar=”^” ; 

 

To apply formatting in-line use the escape character followed by {style[format attributes] } 
all within the quotes. The following formats the first title: 
 

title "^{style[font_face  ='Arial' 

               font_size  =9pt 

               just       =left 

        background =white 

    ] Table  1: (page  ^{thispage} of  ^{lastpage}) ...... 

         }" ; 

 

The following creates a clickable hyperlink in red: 

 
title3 "^{style[font_face  ='Arial' 

                font_size  =7pt 

      font_style =italic 

                just       =left 

      background =white 

     ] Excel and PDF versions (with more data ......): 

              } ^{style[font_face  ='Arial' 

                        font_size  =7pt 

              font_style =italic  

                        color      =Crimson 

                        url='http://www.cdc.gov/.......htm#001' 

                       ] http://www.cdc.gov/........htm#001. 

         }" ; 

CSS & ODS Templates 
Cascading Style Sheets are typically used 
to format HTML documents. This same 
formatting can be used for all ODS 
destinations like PDF, HTML5, and Excel. By 
applying CSS code to an ODS template, the 
CSS overrides selected attributes of the 
template.  

Saving the CSS code in a central location 
for all the Health, United States table 
programs to access allows global changes 
to be made easily. For example, if a 
decision is made to change a font or color 
in a selected attribute for all the Health, 
United States trend tables, the change 
could be made in a global CSS style sheet 
and all reports run after that would use the 
modified style. Figure 6 provides an 
example of CSS formatting style. 

 
Figure 6. CSS Code for Header Cells 

 
One method to apply the CSS styles is to import CSS attributes into an ODS PROC 
TEMPLATE. The CSS attributes will override existing template style attributes, which can 
then be applied as needed: 

 
proc template ; 
  define style styles.HUSHTML5CSSStyle ; 
    parent=Styles.Statistical ; 
      import "&ROOT_FOLDER.\HUS_CSS\HUS_HTML5_StyleSheet.css" ; 
  end ; 
run ; 

 

SAS ODS REPORT WRITING INTERFACE 
The complex format of the Health, United States trend tables requires coding that cannot be 
accomplished in a standard SAS Procedure output and would historically require complex 
DATA steps or PROC REPORT coding. The RWI provides a more elegant method for this type 
of table generation. The RWI is included as a production version in version 9.4 of Base SAS. 
Interim SAS updates (now 9.4 maintenance 6) continue to improve the RWI with additional 
features. The RWI gives full control for report writing. As part of the SAS ODS, users can 
use RWI to take advantage of various formatting including true proportional fonts, styles, 
colors, Unicode characters, traffic lighting, and backgrounds. The RWI allows for exact pixel 
placement of text, images, and custom tables that can be controlled through the DATA step. 

For this project the RWI is the start to finish technology used to create the properly 
formatted trend tables in the various output file types that can include: PDF, HTML5, and 
Excel. The RWI takes an existing process that includes various steps and technology and 
reduces it down to one step taking a SAS dataset and outputting properly formatted trend 
table reports in PDF and HTML5 formats.    

Table Creation using DATA _NULL_ 
DATA _NULL_ is simply a DATA step where there is no output SAS dataset. This is used 
when outputting text files instead of SAS datasets. DATA _NULL_ for report writing has been 
used in the past to include printer control characters to produce custom reports of tabular 
data that would render properly at a physical printer. While the RWI also uses the DATAT 
_NULL_ DATAT step, the RWI provides a robust object-oriented language to provide 
flexibility and full control to create tables with formatting controlled by the DATA step. This 
allows for controls including: conditional logic, formatting capabilities, BY-group processing, 
arrays, and many more.   

 

 
Table 4. NEW Master Table 

Step 1: To control the output of the data table to 
the prescribed custom report, the data must first 
be transposed from the “indicator” style table that 
the data is stored in, referred to as the Master 
table, to one that resembles the desired report. 
Table 4 shows a sample of the formatted Master 
table being transposed to the table format needed 
for the custom output shown in Table 5.  

 

 

 

 
Table 5. Transposed Table for Report  

Figure 7. Sample of Formatted Table Created with SAS RWI 

 

Step 2: The code below initiates the PDF and HTML5 ODS output destination with a few 
options with syntax specific to each file type. 
 
  * PDF FILE * ; 
  ODS PDF file = "&ROOT_FOLDER.\...\HUS_&YEAR._&TABABR._&SUFFIX..pdf"  
          style = styles.HUSPDFCSSStyle 
     startpage = ON       
     title = "PDF Title: Health U.S. &YEAR. Table &TABABR." ; 
 
  * HTML FILE * ; 
  filename filedir "&ROOT_FOLDER.\...\" ; 
  ODS HTML5 path = filedir 
            file = "HUS_&YEAR._&TABABR._&SUFFIX..html"  
  (title = "HTML Title: Health U.S. &YEAR. Table &TABABR.")  
            style = styles.HUSHTML5CSSStyle ; 
 
Step 3: The following code demonstrates how to take the transposed dataset Table 5 and 
create a formatted output similar to the Health, United States table, “Table 1. Resident 
Population, by Age, Sex, Race, and Hispanic Origin: United States, Selected Years 1950–
2016” (Figure 7). Some of the formatting demonstrated below include: nested column 
headers, splitting labels, creating accessibility tags, and various style formatting: 

 
data _NULL_ ; 
  set WORK_DATA end=LAST ; 
    by group_num_001 ; 
    if (_n_=1) then do ;   /* 1 */ 
      dcl odsout HUS() ; 
      HUS.table_start(name: "EXAMPLE_TABLE_001") ; 
        HUS.head_start() ;   /* 2 */  
          HUS.row_start() ;  
            HUS.format_cell() ; 
            HUS.format_cell() ; 
            HUS.format_cell(data:"AGE", column_span:3) ;   /* 3 */ 
          HUS.row_end() ; 
          HUS.row_start() ;  
            HUS.format_cell(data:"Sex, race, Hispanic|origin, and year", 
                            split:"|", vjust:'B') ;   /* 4 */ 
            HUS.format_cell(data:"Total|resident|population", split:"|") ; 
            HUS.format_cell(data:"|Under 1|year"            , split:"|") ; 
            HUS.format_cell(data:"|  1-4|years"             , split:"|") ; 
            HUS.format_cell(data:"| 5-14|years"             , split:"|") ; 
          HUS.row_end() ; 
        HUS.head_end() ;     /* 2 */   
    end ;                  /* 1 */ 

    * PRINT GROUP NAME (Formatted Group Number at top of each GROUP) * ; 
    if (first.group_num_001) then do ;   /* 5 */ 
      HUS.row_start() ;  
        HUS.format_cell(data:group_num_001, type:'H',   /* 6 */ 
                        format:'GROUP_NUM_001F.') ; 
        HUS.format_cell(data:"Number, in thousands", type:'H', 
                        column_span:4) ; 
      HUS.row_end() ; 
    end ;                                /* 5 */  
 
    * OUTPUT DATA WITH ROW HEADERS * ; 
    HUS.row_start() ;   /* 7 */      
      HUS.format_cell(data:put(year_001,4.)||"^{leaders .}",   /* 8 */  
                      type:'H',   /* 9 */ 
                      style_elem:'RowHeader',   /* 10 */  
                      inline_attr:'color=blue',   /* 11 */  
                      just:'left') ; 
      HUS.format_cell(data:group_population,  
                      style_attr:'color=red just=right') ;   /* 12 */ 
      HUS.format_cell(data:age_cat_01, just:'right') ; 
      HUS.format_cell(data:age_cat_02, just:'right') ; 
      HUS.format_cell(data:age_cat_03, just:'right') ; 
    HUS.row_end() ;     /* 7 */ 
 
    if (LAST) then do ;   /* 13 */ 
      HUS.table_end() ; 
      HUS.delete() ;  
    end ;                 /* 13 */ 
run ; 

 
1. The DATA _NULL_ statement starts a data step where no SAS dataset is being 

created.  Comment /* 1 */ marks an IF-THEN code block of what to do for the first 
record of the dataset. In this block an ODSOUT object called “HUS” is declared 
followed by nesting Table, Header, Row, and Cell methods. This is to create the first 
2 rows of the output file column headers.  

2. The HEAD_START() and HEAD_END() methods (comment /* 2 */) identify these 
rows as column headers. Identifying the header rows allows the output PDF and 
HTML5 files to be tagged for accessibility. In these column headers, “AGE” spans 3 
columns of age groups below with the COLUMN_SPAN attribute used in the 
FORMAT_CELL() method (comment /* 3 */). In the second row the labels are too 
long to fit in the desired width of the columns so the SPLIT:”|” attribute identifies the 
Pipe character “|” as the character to mark where the labels should be split in the 
DATA: attribute (comment /* 4 */). 

3. Comment /* 5 */ marks what to do for the first record of each “Sex, race, Hispanic 
origin, and year” group. This is referred to as “BY-Group” processing in SAS. In this 
example the first group is “All Persons”. Like the previous section this needs to be 
identified as a header for accessibility. This case demonstrates the TYPE:’H’ attribute 
of the FORMAT_CELL() method to identify these cells as headers for appropriate 
tagging (comment /* 6 */). 

4. Comment /* 7 */ mark the ROW_START() and ROW_END() methods that surround 
the nested FORMAT_CELL() methods to process the rest of the rows in the dataset. 
For each row the YEAR is displayed with “……” following the YEAR values (comment 
/* 8 */). Comments /* 9 */ and /* 10 */ show how the YEAR cell, for each row, is 
identified as a row header for accessibility tagging; using both the TYPE:’H’ and 
STYLE_ELEM:’RowHeader” attributes of the FORMAT_CELL() method. Comments /* 

11 */ and /* 12 */ show 2 different ways to control the position and color of the data 
values in columns 1 and 2. 

5. Finally, comment /* 13 */ marks the block of code for what to do at the end of the 
dataset once all of the data has been processed. Using the TABLE_END() method to 
close the table and DELETE() method to close out the ODSOUT object called “HUS”.   

 

SECTION 508 ACCESSIBILITY 
For tables to be Section 508 compliant for accessibility, metadata must be present for 
screen reader software to properly present the table to a person with visual impairment. 
The metadata codes, referred to as “tags,” are hidden, when visually reading, but used by 
screen reader software to describe the structure and information in a table. To make a table 
Section 508 compliant, tags must be included to differentiate between row and column 
headers and data cell values. 

When using the RWI to generate tables in this way, the tables are automatically tagged 
appropriately for 508 compliance based on the row and column headers. This automatic 
tagging is turned on by using the system option ACCESSIBLETABLE, available in SAS 9.4 
maintenance release 6 and makes the output tables Section 508 compliant by default. The 
ACCESSIBLECHECK option writes messages to the SAS log if common violations of 
accessibility standards are found such as markup errors that cause screen readers to miss 
content. 

Figure 8. Sample of 
PDF Table, Row, and 
Header Tags 

 
Figure 9. Sample of NEW tagged PDF trend table 

Figure 8 demonstrates the way that metadata tags in PDF 
files, generated automatically using the RWI, can be viewed. 
In this example, Adobe Acrobat Professional is used but any 
software capable of properly listing and modifying tags can 
display this. Figure 9 shows the printed output table where 
these tags are embedded and not visible to the reader. To 
make a table read properly by a screen reader, all the row and 
column headers must be tagged as headers. 

In the case of complex tables with nested row and column 
headers, screen readers must be capable of identifying the 
correct headers for each of the data cells. In Figure 8, the top 
left second data value: 3,147 would be noted as the count 
“Number, in thousands” of “Sex, race, Hispanic origin: All 
Persons” in “1950” “Under the Age of 1”. 

Figure 10 shows the tags that are automatically generated 
when using the RWI to output a properly designed table in a 
HTML5 file. To view the 508 tagging click “Show source” in the 
software being used to view the HTML5 file. 

 
Figure 10. Sample HTML5 Table, Row, and Header Tags 

HEALTH, UNITED STATES NEW TREND TABLE PDF 

 
Figure 11. Sample of New PDF (Sections of Table 1) 

 

The final product can be seen in Figure 11, Table 1 PDF version of Health, United States. 
This table illustrates a file generated directly from SAS that includes: 

• The same table structure as the original PDF 

• Some color and style to demonstrate possibilities 

• Section 508-tagging metadata automated through SAS code (as shown in Figure 8) 

HEALTH, UNITED STATES NEW TREND TABLES HTML5  

 
Figure 12. Sample of New HTML5 (Sections of Table 1) 

 

As seen in Figure 12, the Excel version of Table 1 is a file generated directly from SAS that 
includes:  

• The same table structure as the PDF 

• Some color and style to demonstrate possibilities 

• Section 508-tagging metadata produced with SAS code (as shown in Figure 10) 

CONCLUSION 
This case study illustrates how using the SAS ODS and RWI could automate much of the 
Health, United States production process. Section 508 accessibility features, some of which 
are new in SAS 9.4 maintenance release 6, create more efficient Section 508-compliant 
tables.  

Summary of new outputs: 

• PDF reports with custom formatting, intended for printing, can be generated to 
match the existing PDF output, thus maintaining consistency going forward while 
improving the accessibility of the files 

• HTML5 version of these reports can replace the existing Excel files, intended for 
screen reading, with a format that is better suited for online publishing. 

• CSV file would be produced to provide an open format file with the data in a plain 
text format for more direct access to the data for various analytic uses. 

